
Subject: Calculating Timing of First PNC check up from EDHS 2011

Posted by [Hanu](#) on Fri, 08 Jul 2016 14:02:16 GMT

[View Forum Message](#) <> [Reply to Message](#)

Hello,

I am trying to replicate table 9.9 from EDHS2011, particularly looking at the timing of first PNC check up for the mother, in the categories that are present in the report. These are "Less than 4 hours, 4-23 hours, 1-2 days, 3-6 days and 7-41 days". I wanted to ask how generate those categories from the m71 variable in the birth recode file.

Thank you.

Subject: Re: Calculating Timing of First PNC check up from EDHS 2011

Posted by [Bridgette-DHS](#) on Mon, 11 Jul 2016 15:14:04 GMT

[View Forum Message](#) <> [Reply to Message](#)

Following is a response from Senior DHS Stata Specialist, Tom Pullum:

Don't confuse m51 and m71. m51 refers to the mother's checkup and m71 to the child's checkup. m51 is the variable used in table 9.9.

The following lines will give a match with the indicator in table 9.9 (for the total sample). m51 is a three-column code, with 1 in the first column for hours, 2 in the first column for days, 3 in the first column for weeks. Columns 2-3 give the completed number of units. For example, m51=305 is five completed weeks, which is equivalent to 41 completed days. To understand this variable it helps to list its distribution and its label, which is also called me51.

```
set more off
label list m51
tab m51 if v008-b3<24 & bidx==1,m
```

```
gen time=.
replace time=1 if m51<104
replace time=2 if m51>=104 & m51<=123
replace time=3 if m51>=124 & m51<=202
replace time=4 if m51>=203 & m51<=206
replace time=5 if (m51>=207 & m51<=241) | (m51>=301 & m51<=305)
replace time=8 if m51==199 | m51==299 | m51==399 | m51==998 | m51==999
replace time=9 if time==.
```

```
label define time 1 "<4 hours" 2 "4-23 hours" 3 "1-2 days" 4 "3-6 days" 5 "7-41 days" 8 "dk" 9
"none"
```

```
label values time time
```

```
label variable time "Woman's time to postnatal care"
```

tab time
tab time if v008-b3<24 & bidx==1 [iweight=v005/1000000]

Subject: Re: Calculating Timing of First PNC check up from EDHS 2011
Posted by [umeshg](#) on Sun, 06 May 2018 09:12:33 GMT
[View Forum Message](#) <> [Reply to Message](#)

Dear DHS,
I used this code to calculate the PNC services for 2011 NDHS. The result, when I compared with the report and also with Statcompiler is .5 less than what it calculates (44.5 vs 44.6). Code looks fine but I could not figure out the mistake ?

Thanks

Subject: Re: Calculating Timing of First PNC check up from EDHS 2011
Posted by [Bridgette-DHS](#) on Mon, 07 May 2018 12:55:21 GMT
[View Forum Message](#) <> [Reply to Message](#)

You have posted this under a discussion on Ethiopia 2011 DHS, but are referencing NDHS 2011. Please clarify which country's dataset you are working with and also specify the number of the table you are trying to replicate.

Subject: Re: Calculating Timing of First PNC check up from EDHS 2011
Posted by [umeshg](#) on Mon, 07 May 2018 14:24:51 GMT
[View Forum Message](#) <> [Reply to Message](#)

Dear Bridgette,
I am referring to Table 9.10 of Nepal DHS report. Percentage of PNC checkup in the first two days after delivery is 44.5 while I got 44.6 %.

Thanks

Subject: Re: Calculating Timing of First PNC check up from EDHS 2011
Posted by [Bridgette-DHS](#) on Mon, 07 May 2018 17:36:13 GMT
[View Forum Message](#) <> [Reply to Message](#)

Following is a response from DHS Senior Research Associate, Lindsay Mallick:

Although the table does not specify, the column "Percentage of women with a postnatal checkup in the first two days after birth" is based on whether the woman had PNC in the first 2 days and

the check was performed by a skilled provider. You can recreate the table with the following code:

```
use NPIR60FL.dta, clear
```

```
*most recent child born in last two years  
gen b19 = v008 - b3_01
```

```
*timing of pnc for most recent birth  
recode m51_1 ( 100/202 =2 "PNC within 2 days") (203/305= 1 "PNC 3-41 days") (900/1000 =9  
"Don't know, missing") (else=0 "No PNC or after 41 days") if b19<24, gen(pncntime)
```

```
*tabulate with weights  
ta pncntime [iw=v005/1000000]
```

```
*pnc in 2 days by a skilled provider  
gen pnc2d = 0 if b19<24  
replace pnc2d = 1 if pncntime==2 & m52_1 <30 & b19<24
```

```
*tabulate  
ta pnc2d [iw=v005/1000000]
```

Subject: Re: Calculating Timing of First PNC check up from EDHS 2011

Posted by [umeshg](#) on Tue, 08 May 2018 06:27:16 GMT

[View Forum Message](#) <> [Reply to Message](#)

Thank you very much Lindsay Mallick and Bridgette
